Stata Spain Conference

Introduction to Bayesian VAR models in Stata

Gustavo Sánchez

StataCorp LLC

October 20, 2022
Madrid, Spain

Outline

1. Bayesian analysis: Brief overview
   - The general idea
   - The method
2. The Stata tools for Bayesian analysis
3. Bayesian linear regression
4. VAR model
5. Bayesian VAR
   - Estimation
   - Postestimation

The method

The general idea

Frequentist

Data hypothetically repeatable

Theoretical Model →

```
. list month defunciones casos_uci, abbreviate(12)
```

	month	defunciones	casos_uci
1.	2021m11	631	524
2.	2021m12	1912	1298
3.	2022m1	5453	1740
4.	2022m2	4183	691
5.	2022m3	1600	382
6.	2022m4	1422	436
7.	2022m5	1848	620
8.	2022m6	1663	601
9.	2022m7	3133	696
10.	2022m8	1046	219

→



Bayesian

Data known

Theoretical Model →

```
. list month defunciones casos_uci, abbreviate(12)
```

	month	defunciones	casos_uci
1.	2021m11	631	524
2.	2021m12	1912	1298
3.	2022m1	5453	1740
4.	2022m2	4183	691
5.	2022m3	1600	382
6.	2022m4	1422	436
7.	2022m5	1848	620
8.	2022m6	1663	601
9.	2022m7	3133	696
10.	2022m8	1046	219

→





The method

- Inverse law of probability (Bayes' Theorem):

$$p(\theta|y) = \frac{p(y|\theta)\,p(\theta)}{p(y)} = \frac{f(y;\theta)\,\pi(\theta)}{f(y)}$$

Where:

$f(y;\theta)$: probability density function for y given θ.

$\pi(\theta)$: prior distribution for θ

- The marginal distribution of y, f(y), does not depend on θ; then we can write the fundamental equation for Bayesian analysis:

$$p(\theta|y) \propto L(\theta;y)\,\pi(\theta)$$

Where:

$L(\theta;y)$: likelihood function of the parameters given the data.

The method

- Some prior-likelihood combinations have closed form solution.
- What about the cases with non-closed solutions, or more complex distributions?
  - Integration is performed via simulation.
  - We need to use intensive computational simulation tools to find the posterior distribution in most cases.
  - Markov chain Monte Carlo (MCMC) methods are the current standard in most software. Stata implements two alternatives:
    - Metropolis–Hastings (MH) algorithm
    - Gibbs sampling

The method

- Links for Bayesian analysis and MCMC on our YouTube channel:
    - Introduction to Bayesian statistics, part 1: The basic concepts

      https://www.youtube.com/watch?v=0F0QoMCSKJ4&feature=youtu.be

    - Introduction to Bayesian statistics, part 2: MCMC and the Metropolis–Hastings algorithm.

      https://www.youtube.com/watch?v=OTO1DygELpY&feature=youtu.be

The method

- Metropolis–Hastings simulation
  - The trace plot illustrates the sequence of accepted proposal states for a simulation with enough burnin iterations.



The method

- We expect to obtain a stationary sequence when convergence is achieved.



The method

- An efficient MCMC should have small autocorrelation.
- We expect autocorrelation to become negligible after a few lags.



The Stata tools for Bayesian regression

Stata's convenient syntax: `bayes:`

```
regress y x1 x2 x3

bayes: regress y x1 x2 x3


logit y x1 x2 x3

bayes: logit y x1 x2 x3


mixed y x1 x2 x3 || region:

bayes: mixed y x1 x2 x3 || region:
```

Stata's Bayesian suite consists of the following commands

Command	*Description*
Estimation	
`bayes:`	Bayesian regression models using the `bayes` prefix
`bayesmh`	General Bayesian models using MH
`bayesmh` *evaluators*	User-defined Bayesian models using MH
Postestimation	
`bayesgraph`	Graphical convergence diagnostics
`bayesstats ess`	Effective sample sizes and more
`bayesstats grubin`	Gelman–Rubin convergence diagnostics
`bayesstats summary`	Summary statistics
`bayesstats ic`	Information criteria and Bayes factors
`bayestest model`	Model posterior probabilities
`bayestest interval`	Interval hypothesis testing
`bayespredict`	Bayesian predictions (available only after `bayesmh`)
`bayesstats ppvalues`	Bayesian predictive *p*-values (available only after `bayesmh`)
Added in latest version	
`bayes:var`	Bayesian VAR models
`bayes:dsge`	Bayesian DSGE models
`bayes:xt`	Bayesian panel data models

Example 1: Covid mortality in Spain

- Let's work with a simple linear regression for the covid_19 mortality in Spain as a function of the number of hospitalizations and a linear trend:
  - We used data from the Spanish National Center for Epidemiology.
  - Let's consider the following model specification:

$$\textit{diseased} = \alpha_1 + \beta_{\textit{hospitalized}} * \textit{hospitalized} + \beta_{\textit{trend}} * \textit{trend} + \epsilon_1$$

Where:

diseased	:	Monthly number of deseased due to Covid-19.
hospitalized	:	Monthly number of hospitalized due to Covid-19.
trend	:	Linear trend.

Data

- Spanish Data for Covid-19 data 2020m1-2022m8:

```
. describe

Contains data from C:\Users\gas\Documents\conferences\Spain\2022\data\covid_merged.dta
 Observations:            32                  Covid-19 data for Spain 2020m1-2022m8
    Variables:             4                  15 Sep 2022 11:20
--------------------------------------------------------------------------------------
Variable      Storage   Display    Value
    name         type    format    label      Variable label
--------------------------------------------------------------------------------------
month           float   %tm
hospitalized    double  %10.0g                Hospitalizations due to Covid-19
diseased        double  %10.0g                Diseased due to Covid-19
trend           float   %9.0g
--------------------------------------------------------------------------------------
Sorted by: month
     Note: Dataset has changed since last saved.
```



Linear Regression

- Linear regression with the `bayes:` prefix

```
. bayes,rseed(1):regress diseased hospitalized trend
```

- Equivalent model with `bayesmh`

```
. bayesmh diseased hospitalized trend, rseed(1)      ///
>     likelihood(normal({sigma2}))                   ///
>     prior({diseased:hospitalized}, normal(0,10000)) ///
>     prior({diseased:trend},        normal(0,10000)) ///
>     prior({diseased:_cons},        normal(0,10000)) ///
>     prior({sigma2}, igamma(.01,.01))                ///
>     block({diseased:hospitalized trend _cons})      ///
>     block({sigma2})
```

bayes: prefix

```
. bayes,rseed(1):regress diseased hospitalized trend

Burn-in ...
Simulation ...
Model summary
------------------------------------------------------------------------------
Likelihood:
  diseased ~ regress(xb_diseased,{sigma2})
Priors:
  {diseased:hospitalized trend _cons} ~ normal(0,10000)                    (1)
                             {sigma2} ~ igamma(.01,.01)
------------------------------------------------------------------------------
(1) Parameters are elements of the linear form xb_diseased.
Bayesian linear regression                       MCMC iterations  =     12,500
Random-walk Metropolis-Hastings sampling         Burn-in          =      2,500
                                                 MCMC sample size =     10,000
                                                 Number of obs    =         32
                                                 Acceptance rate  =      .3784
                                                 Efficiency:  min =     .06949
                                                              avg =     .09369
Log marginal-likelihood = -302.64048                          max =      .1197

------------------------------------------------------------------------------
             |                                                Equal-tailed
             |      Mean   Std. dev.     MCSE     Median  [95% cred. interval]
-------------+----------------------------------------------------------------
diseased     |
hospitalized |  .2923772   .0275787   .001046    .292421   .2376055   .3454921
       trend | -38.42549   24.50846   .852808  -38.71426  -86.56636   10.16956
       _cons |  20.40253    98.6312   3.07287   20.47222  -171.0139   216.2227
-------------+----------------------------------------------------------------
      sigma2 |   4278154    1133774     32777    4114782    2545187    6988581
------------------------------------------------------------------------------
Note: Default priors are used for model parameters.
Note: Adaptation tolerance is not met in at least one of the blocks.
```

bayesgraph

- We can use `bayesgraph` to look at the trace, the correlation, and the density. For example:

```
. bayesgraph diagnostic {hospitalized}
```



- The trace indicates that convergence was achieved
- Correlation becomes negligible after 15 periods

bayesgraph

- We can use `bayesgraph` to look at the trace, the correlation, and the density. For example: inflation

```
. bayesgraph diagnostic {trend}
```



- The trace also indicates that convergence was achieved
- Correlation becomes negligible after 15 periods

bayesgraph

- We can use `bayesgraph` to look at the trace, the correlation, and the density. For example:

```
. bayesgraph diagnostic {sigma2}
```



- The trace indicates that convergence was achieved
- Correlation becomes negligible after 10 periods

Vector Autoregressive (VAR) Models

Probabilities available with Bayesian VARs

- Can we compute probabilities for events associated to multiple equation forecasts?

```
. collect preview
```

	2010Q1	2010Q1	2010Q1-Q2
Event			
inflation_over_5=1	.6778		
infl_over_5__exchrate_chg_over_3=1		.4554	
exchrate_change_over_3=1			.4303

- Probability forecasting (Garrat et al. (2006)) allows defining events for forecasted variables conditional in the estimation sample.
  - Forecasts are based on econometric models subject to uncertainty on the future, on the parameters, on the model, and also on the policies.
  - See an example in Sanchez and Zavarce (2013) for probability forecast accounting for future uncertainty.
- The Bayesian approach allows obtaining probabilities for events based on parameters and future uncertainty.

Probabilities available with Bayesian VARs

- Can we compute probabilities for events associated to multiple equation forecasts?

```
. collect preview
```

	2010Q1	2010Q1	2010Q1-Q2
Event			
inflation_over_5=1	.6778		
infl_over_5__exchrate_chg_over_3=1		.4554	
exchrate_change_over_3=1			.4303

- Probability forecasting (Garrat et al. (2006)) allows defining events for forecasted variables conditional in the estimation sample.
  - Forecasts are based on econometric models subject to uncertainty on the future, on the parameters, on the model, and also on the policies.
  - See an example in Sanchez and Zavarce (2013) for probability forecast accounting for future uncertainty.
- The Bayesian approach allows obtaining probabilities for events based on parameters and future uncertainty.

Vector Autoregressive Models VAR

- VARs are extensions of AR(p) models for vector valued dependent variables with no structural form.
- A VAR model can be written as:

$$\mathbf{y_t} = \mathbf{v} + \mathbf{A_1 y_{t-1}} + \ldots + \mathbf{A_p y_{t-p}} + \mathbf{C_0 x_t} + \mathbf{C_1 x_{t-1}} + \ldots + \mathbf{C_s x_{t-s}} + \mathbf{u_t}$$

Where:

$\mathbf{y}_t = (y_{1t}, ..., y_{Kt})$ is a Kx1 random vector
$\mathbf{A_1}$ through $\mathbf{A_p}$ are KxK matrices of parameters.
$\mathbf{x_t}$ is an Mx1 vector of exogenous variables
$\mathbf{C_0}$ through $\mathbf{C_s}$ are KxM matrices of parameters.
$\mathbf{v}$ is a Kx1 vector of parameters
$\mathbf{u_t}$ is a vector assumed to be white noise:

$$\begin{aligned} E(\mathbf{u_t}) &= \mathbf{0} \\ E(\mathbf{u_t u_t}') &= \mathbf{\Sigma} \\ E(\mathbf{u_t u_s}') &= \mathbf{0} \;;\; t \neq s \end{aligned}$$

- The number of coefficients is quadratic to the number of dependent variables and proportional the number of lags.

Example 2: VAR model with Covid-19 data. Estimation with the `var` command

```
. var diseased hospitalized,exog(trend) lags(1/2) vsquish
Vector autoregression
Sample: 2020m3 thru 2022m8                     Number of obs     =         30
Log likelihood =  -561.5529                    AIC               =   38.23686
FPE            =   1.40e+14                    HQIC              =   38.41616
Det(Sigma_ml)  =   6.22e+13                    SBIC              =   38.79734
Equation           Parms      RMSE     R-sq      chi2     P>chi2
----------------------------------------------------------------
diseased             6     2377.63   0.6463   54.82962   0.0000
hospitalized         6       10433   0.2635    10.7308   0.0570
----------------------------------------------------------------

------------------------------------------------------------------------------
             | Coefficient  Std. err.      z    P>|z|     [95% conf. interval]
-------------+----------------------------------------------------------------
diseased     |
    diseased |
         L1. |    -.64971   .4223594    -1.54   0.124    -1.477519    .1780991
         L2. |  -.0947683   .2295219    -0.41   0.680    -.5446229    .3550863
hospitalized |
         L1. |   .3557982   .0953376     3.73   0.000       .16894    .5426564
         L2. |   .0354961   .1080252     0.33   0.742    -.1762294    .2472216
       trend |  -242.9167   72.40319    -3.36   0.001    -384.8244   -101.0091
       _cons |   5253.857   1281.884     4.10   0.000      2741.41    7766.303
-------------+----------------------------------------------------------------
hospitalized |
    diseased |
         L1. |  -3.834202   1.853307    -2.07   0.039    -7.466617   -.2017865
         L2. |  -1.000171   1.007139    -0.99   0.321    -2.974126    .9737849
hospitalized |
         L1. |   1.072752   .4183399     2.56   0.010     .2528208    1.892683
         L2. |   .5993708    .474013     1.26   0.206    -.3296776    1.528419
       trend |  -733.3267   317.7042    -2.31   0.021    -1356.016   -110.6379
       _cons |   21402.79   5624.888     3.81   0.000     10378.21    32427.36
------------------------------------------------------------------------------
```

Bayesian VAR models with `bayes:var`

- Overparameterization in VAR models is particular problematic with small samples.
- Bayesian VAR allows shrinking the vector of regression coefficients by controlling the effective number of lags through the priors.
- The Minnesota family of priors represent a flexible specification that allows the expert's knowledge to be incorporated in the estimation.
- Bayes factors can be used to select the number of lags, and also the exogenous variables.

Bayesian VAR models with `bayes:var`

- The Bayesian approach to fit VAR models assigns prior distributions to all the regression parameters:
  - The likelihood is derived from the linear specification

$$\mathbf{y_t} = \mathbf{A_1 y_{t-1}} + \ldots + \mathbf{A_p y_{t-p}} + \mathbf{C x_t} + \mathbf{u_t} \quad ; \quad \mathbf{u_t} \sim N(0, \Sigma)$$

  - For the regression coefficients $\beta = vec(\mathbf{C}, \mathbf{A_1}, ..., \mathbf{A_p})$ the prior corresponds to a multivariate normal:

$$\beta | \boldsymbol{y} \sim N\left(\boldsymbol{\beta_0}, \boldsymbol{\Omega}\right)$$

  - For the regression covariance matrix $\boldsymbol{\Sigma}$ the prior distribution would be either inverse Wishart or Jeffreys.

Minnesota priors

- `bayes:var` has four prior family alternatives:
  - Minnesota prior with fixed covariance $\boldsymbol{\Sigma}$

    `bayes, minnfixedcovprior... : var...`

  - Conjugate Minnesota prior (The default)

    `bayes, minnconjprior... : var...`

  - Minnesota prior for β and inverse-Wishart prior for $\boldsymbol{\Sigma}$

    `bayes, minnwishprior... : var...`

  - Minnesota prior for β and Jeffreys prior for $\boldsymbol{\Sigma}$

    `bayes, minnjeffprior... : var...`

Original Minnesota prior with fixed covariance

- Doan, Litterman, and Sims 1984 and Litterman 1986, assumed a known fixed-error covariance matrix.
- Prior for vector of coefficients:

$$\beta \sim N(\beta_{\mathbf{0}}, \boldsymbol{\Omega}_{\mathbf{0}})$$

- $\boldsymbol{\Omega}_{\mathbf{0}}$ is diagonal (i.e. no correlation between the coefficients in β).

- The covariance matrix of the error is known ($\boldsymbol{\Sigma} = \boldsymbol{\Sigma}_{\mathbf{0}}$):

$$\mathbf{u} \sim N(\mathbf{0}, \boldsymbol{\Sigma}_{\mathbf{0}} \otimes \boldsymbol{I}_T)$$

Original Minnesota prior with fixed covariance

- Example: $\mathbf{y_{1t}} = a_{11} y_{1t-1} + a_{12} y_{2t-1} + c1 + u_{1t}$
  $\mathbf{y_{2t}} = a_{21} y_{1t-1} + a_{22} y_{2t-1} + c2 + u_{1t}$

- Independent error terms with known fixed variances

$$u_1 \sim N\left(0, \hat{\sigma}_1^2\right),\ u_2 \sim N\left(0, \hat{\sigma}_2^2\right)$$

- Prior expectations: $E[a_{11}] = E[a_{22}] = 1$
  $E[a_{12}] = E[a_{21}] = E[c_1] = E[c_2] = 0$

- Prior variances:
$$\begin{aligned} Var[a_{11}] &= Var[a_{22}] = \lambda_1^2 \\ Var[a_{12}] &= Var[a_{21}] \sim \lambda_1^2 \lambda_2^2 \\ Var[c_1] &= Var[c_2] \sim \lambda_1^2 \lambda_4^2 \end{aligned}$$

  Assuming $\lambda_3 = 1$

`bayes:var` with fixed covariance

- Default estimation with `minnfixedcovprior`. Original Minnesota prior with $\lambda_1 = 0.1$, $\lambda_2 = 0.5$, $\lambda_4 = 100$

```
bayes, minnfixedcovprior:var y1 y2,lags(1)
```

- Increase the self-tightness, for example, set $\lambda_1 = 1$

```
bayes, minnfixedcovprior(selftight(1)):    ///
       var y1 y2,lags(1)
```

- Specify zero-mean priors for all the coefficients:

```
bayes, minnfixedcovprior(mean(0,0)):       ///
       var y1 y2,lags(1)
```

- Reduce the exogenous-variables tightness parameter, for example, set it to $\lambda_4 = 50$

```
bayes, minnfixedcovprior(exogtight(50)):   ///
       var y1 y2,lags(1)
```

Example 3: Bayesian VAR estimation with Covid-19 data
Default values for fixed-error covariance

```
 . matrix b0 = J(1,2,0)
. bayes,minnfixedcovprior(mean(b0)) rseed(123) dryrun: ///
>   var diseased hospitalized,exog(trend) lags(1/2)
Model summary
------------------------------------------------------------------------------
Likelihood:
  diseased hospitalized ~ mvnormal(2,xb_diseased,xb_hospitalized,_Sigma0)
Priors:
          {diseased:L(1 2).diseased}                                       (1)
      {diseased:L(1 2).hospitalized}                                       (1)
                    {diseased:trend}                                       (1)
                    {diseased:_cons}                                       (1)
      {hospitalized:L(1 2).diseased}                                       (2)
  {hospitalized:L(1 2).hospitalized}                                       (2)
                {hospitalized:trend}                                       (2)
                {hospitalized:_cons} ~ minnesota(2,2,2,b0,_Sigma0,.1,.5,1,100) (2)
------------------------------------------------------------------------------
(1) Parameters are elements of the linear form xb_diseased.
(2) Parameters are elements of the linear form xb_hospitalized.
```

Default values for fixed-error covariance

```
. matrix b0 = J(1,2,0)
. bayes,minnfixedcovprior(mean(b0)) rseed(321) noheader nomodelsummary: ///
>     var diseased hospitalized,exog(trend) lags(1/2)

Burn-in ...
Simulation ...
```

	Mean	Std. dev.	MCSE	Median	Equal-tailed [95% cred.	interval]
diseased						
diseased						
L1.	.0947181	.086135	.000861	.0945705	-.0754878	.2630525
L2.	-.021198	.0474565	.000475	-.0208138	-.1147206	.071278
hospitalized						
L1.	.0138442	.0141551	.000139	.0137927	-.0141889	.0414014
L2.	-.0008147	.0071874	.000073	-.0008389	-.0148412	.0131862
trend	-166.2745	63.3547	.633547	-166.361	-290.4174	-41.89737
_cons	6018.406	1342.198	13.2686	6048.237	3376.494	8633.275
hospitalized						
diseased						
L1.	-.0094429	.1649367	.001667	-.0097426	-.3290775	.3106183
L2.	-.0186537	.0858377	.000858	-.0184087	-.1886121	.1508307
hospitalized						
L1.	.0626832	.0872694	.000873	.06309	-.1085091	.2308297
L2.	-.0143528	.0479871	.00048	-.0139387	-.1100908	.0775663
trend	-222.518	214.6897	2.10931	-220.5278	-643.418	195.0401
_cons	16721.54	4485.967	44.2321	16693.98	8018.407	25535.59

Conjugate Minnesota prior (Default for `bayes:var`)

- Example: $$\begin{aligned} \mathbf{y_{1t}} &= a_{11}y_{1t-1} + a_{12}y_{2t-1} + c1 + u_{1t} \\ \mathbf{y_{2t}} &= a_{21}y_{1t-1} + a_{22}y_{2t-1} + c2 + u_{1t} \end{aligned}$$

- Error terms have a multivariate normal distribution

$$(\mathbf{u_1}, \mathbf{u_2}) \sim MVN\left((0,0), \boldsymbol{\Sigma}\right)$$

- Prior expectations: $$\begin{aligned} E[a_{11}] &= E[a_{22}] = 1 \\ E[a_{12}] &= E[a_{21}] = E[c_1] = E[c_2] = 0 \end{aligned}$$

- Prior variances: $Var[a_{11}, a_{12}, a_{21}, a_{22}, c_1, c_2] = \boldsymbol{\Sigma} \otimes \boldsymbol{\Phi_0}$

$$\boldsymbol{\Sigma} \sim InvWishart\left(\alpha_0, \boldsymbol{S_0}\right) \quad ; \quad \begin{aligned} \alpha_0 &= K + 2 \\ \boldsymbol{S_0} &= (\alpha_0 - K - 1)\boldsymbol{\Sigma_0} \end{aligned}$$

$$\boldsymbol{\Phi_0} = diag\left(\frac{\lambda_1^2}{\hat{\sigma}_1^2}, \frac{\lambda_1^2}{\hat{\sigma}_2^2}, \lambda_1^2\lambda_4^2\right)$$

bayes:var with conjugate Minnesota prior

- Default estimation with minnconjprior. Original Minnesota prior with $\lambda_1 = 0.1$ and $\lambda_4 = 100$

```
bayes, minnconjprior:var y1 y2,lags(1)
```

- Increase the self-tightness, for example, set $\lambda_1 = 1$

```
bayes, minnconjprior(selftight(1)):     ///
       var y1 y2,lags(1)
```

- Specify zero-mean priors for all the coefficients:

```
bayes, minnconjprior(mean(0,0)):        ///
       var y1 y2,lags(1)
```

- Reduce the exogenous-variables tightness parameter, for example, set it to $\lambda_4 = 50$

```
bayes, minnconjprior(exogtight(50)):  ///
       var y1 y2,lags(1)
```

Conjugate prior with self-tightness equal to 1

```
 . matrix b0 = J(1,2,0)
. bayes,minnconjprior(mean(b0) selftight(1)) rseed(123) dryrun: ///
>      var diseased hospitalized,exog(trend) lags(1/2)
Model summary
------------------------------------------------------------------------------

Likelihood:
  diseased hospitalized ~ mvnormal(2,xb_diseased,xb_hospitalized,{Sigma,m})
Priors:
          {diseased:L(1 2).diseased}                                        (1)
      {diseased:L(1 2).hospitalized}                                        (1)
                   {diseased:trend}                                         (1)
                   {diseased:_cons}                                         (1)
      {hospitalized:L(1 2).diseased}                                        (2)
  {hospitalized:L(1 2).hospitalized}                                        (2)
               {hospitalized:trend}                                         (2)
               {hospitalized:_cons} ~ varconjugate(2,2,2,b0,{Sigma,m},_Phi0) (2)
                          {Sigma,m} ~ iwishart(2,4,_Sigma0)
------------------------------------------------------------------------------

(1) Parameters are elements of the linear form xb_diseased.
(2) Parameters are elements of the linear form xb_hospitalized.
```

Conjugate prior / self-tightness equal to 1

```
 . matrix b0 = J(1,2,0)
. bayes,minnconjprior(mean(b0) selftight(1))      ///
>     noheader nomodelsummary rseed(123):         ///
>     var diseased hospitalized,exog(trend) lags(1/2)

Burn-in ...
Simulation ...
```

	Mean	Std. dev.	MCSE	Median	Equal-tailed [95% cred.	interval]
diseased						
diseased						
L1.	-.329615	.3333668	.003334	-.3293344	-.9838306	.3307015
L2.	-.0648119	.1820188	.001853	-.0648933	-.4238184	.2941523
hospitalized						
L1.	.2801985	.0812233	.000812	.2800759	.118405	.4406346
L2.	-.0165996	.0739198	.000739	-.0168805	-.1638307	.1270908
trend	-206.0003	63.49083	.634908	-205.8454	-331.4585	-81.57799
_cons	5043.442	1293.023	12.9302	5029.928	2516.001	7615.631
hospitalized						
diseased						
L1.	-2.000969	1.414351	.014144	-2.006827	-4.797872	.7852064
L2.	-.4269968	.7879028	.008027	-.4196723	-2.006221	1.103534
hospitalized						
L1.	.6997713	.3478424	.003478	.6981416	.0118825	1.389073
L2.	.1496314	.3146749	.003147	.1525077	-.4797339	.7654897
trend	-480.6253	272.7801	2.7278	-479.3816	-1020.583	54.52934
_cons	19210.17	5563.18	56.2509	19156.43	8179.651	30246.46
Sigma_1_1	5311427	1386084	13860.8	5109926	3255233	8597325
Sigma_2_1	1.89e+07	5389532	53895.3	1.81e+07	1.07e+07	3.18e+07
Sigma_2_2	9.69e+07	2.51e+07	251401	9.30e+07	5.96e+07	1.57e+08

MVN inverse-Wishart and MVN Jeffreys priors

- Example: $$\mathbf{y_{1t}} = a_{11}y_{1t-1} + a_{12}y_{2t-1} + c1 + u_{1t}$$ $$\mathbf{y_{2t}} = a_{21}y_{1t-1} + a_{22}y_{2t-1} + c2 + u_{1t}$$
- Error terms have a multivariate normal distribution

$$(\mathbf{u_1}, \mathbf{u_2}) \sim MVN\left((0,0), \boldsymbol{\Sigma}\right)$$

- Prior for coefficients:

$$\beta \sim N(\boldsymbol{\beta_0}, \boldsymbol{\Omega_0})$$ where: $\boldsymbol{\beta_0}$ and $\boldsymbol{\Omega_0}$ are those from the original Minnesota prior.

- Prior variances:
  - For MVN inverse Wishart (`minninvwishprior`):

$$\boldsymbol{\Sigma} \sim InvWishart(\alpha_0, \boldsymbol{S_0}) \quad ; \quad \alpha_0 = K + 2, \quad \boldsymbol{S_0} = (\alpha_0 - K - 1)\boldsymbol{\Sigma_0}$$

  - For MVN Jeffreys (`minnjeffprior`): $\boldsymbol{\Sigma} \sim Jeffreys(K)$

Stability condition and Impulse-response functions (IRF)

- If the VAR is stable, we can derive the alternative moving average representation

$$\mathbf{y_t} = \boldsymbol{\mu} + \sum_{i=0}^{\infty} \mathbf{D_i x_{t-i}} + \sum_{i=0}^{\infty} \boldsymbol{\Phi}_i \mathbf{u_{t-1}}$$

Where:

$\mathbf{y_t}$: $I(1)$

$\boldsymbol{\mu}$: Kx1 time-invariant mean of the process

$\boldsymbol{\Phi}_i$: KxK matrices of parameters (MA coefficients - IRF)

$\mathbf{u_{t-1}}, \mathbf{u_{t-2}}, \ldots$: i.i.d shocks

- The **stability condition** is satisfied if all eigenvalues for a companion matrix are less than 1.
- The companion matrix can be derived from the moving average representation.

Orthogonal shocks

- Alternative representation in terms of orthogonal shocks for the IRFs.
- Let's have a matrix $\mathbf{P}$ such that $\mathbf{\Sigma} = \mathbf{P}\mathbf{P}'$

$$
\begin{aligned}
\mathbf{y_t} &= \mu + \sum_{s=0}^{\infty} \mathbf{\Phi}_s \mathbf{P}\mathbf{P}' \mathbf{u_{t-s}} \\
&= \mu + \sum_{s=0}^{\infty} \mathbf{\Theta}_s \mathbf{P}^{-1} \mathbf{u_{t-s}} \\
&= \mu + \sum_{s=0}^{\infty} \mathbf{\Theta}_s \mathbf{w_{t-s}}
\end{aligned}
$$

Where: $E[\mathbf{P}^{-1}\mathbf{u_{t-s}}] = 0$

$$E[\mathbf{P}^{-1}\mathbf{u_t}(\mathbf{P}^{-1}\mathbf{u_t})'] = \mathbf{I_K}$$

Check stability condition

```
. matrix b0 = J(1,2,0)
. quietly bayes,minnconjprior(mean(b0) selftight(1))     ///
>     saving("$simul_dir\bvar_sim1",replace) rseed(123): ///
>     var diseased hospitalized,exog(trend) lags(1/2)
. bayesvarstable
Eigenvalue stability condition                  Companion matrix size =     4
                                                MCMC sample size      = 10000
```

Eigenvalue modulus	Mean	Std. dev.	MCSE	Median	Equal-tailed [95% cred.	interval]
1	.62096	.1156883	.001157	.6234019	.389435	.8404553
2	.5976153	.1234386	.001234	.6029557	.347319	.825098
3	.4046902	.1381563	.001382	.4078005	.1296683	.6630941
4	.2220758	.1346913	.001347	.2128812	.0110299	.4953334

```
Pr(eigenvalues lie inside the unit circle) = 0.9993
```

- The probability that all the eigenvalues are less than one supports the stability of the selected model.

Impulse-response functions - Table

- Impulse-response functions considering the order hospitalized->diseased.

```
 . quietly bayes,minnconjprior(mean(b0) selftight(1))    ///
>     saving("$simul_dir\bvar_sim1",replace) rseed(123): ///
>     var hospitalized diseased,exog(trend) lags(1/2)
. estimates store my_irf
. bayesirf create mybirf, step(8) set(mybirf) replace
(file mybirf.irf now active)
(file mybirf.irf updated)
. bayesirf table oirf,impulse(hospitalized) response(diseased)
Results from mybirf
```

Step	(1) oirf	(1) Lower	(1) Upper
0	1900.44	1336.74	2624.47
1	2107.53	1207.63	3173.54
2	-67.0811	-1101.12	1102.47
3	-477.858	-1586.26	529.208
4	-95.1214	-1017.47	719.198
5	172.012	-316.631	894.777
6	100.918	-291.833	730.643
7	5.87918	-408.49	433.156
8	-25.1055	-404.101	250.999

```
Posterior means reported.
95% equal-tailed credible lower and upper bounds reported.
(1) irfname = mybirf, impulse = hospitalized, and response = diseased.
```

Impulse-response functions - Graph



Example 4: VAR model for prices (CPI) and exchange rate for Spain

Data

- We used `import fred` to get data from the Federal Reserve Economic Data (FRED).

```
 . import fred EXUSEU ESPCPIALLMINMEI WTISPLC,              ///
>          daterange(2010-01-01 2022-08-31) aggregate(monthly,eop)
. generate m=month(daten)
. generate y=year(daten)
. generate month=ym(y,m)
. format %tm month
. tsset month
. tsappend,add(4)
. rename          EXUSEU dollar_euro
. rename ESPCPIALLMINMEI cpi_spain
. rename         WTISPLC oil_price
. generate ldollar_euro   = ln(dollar_euro)
. generate lcpi_spain     = ln(cpi_spain)
. generate loil_price     = ln(oil_price)
```

`import fred`: Dialog box



Data for exchange rate, cpi, and oil price

```
 . describe
Contains data from C:\Users\gas\Documents\conferences\Spain\2022\data\cpi_spain
> _interannual.dta
 Observations:           216
    Variables:             9                  11 Oct 2022 10:27
--------------------------------------------------------------------------------
Variable       Storage   Display    Value
    name         type    format     label      Variable label
--------------------------------------------------------------------------------
datestr         str10    %-10s                 observation date
daten           int      %td                   numeric (daily) date
dollar_euro     float    %9.0g                 U.S. Dollars to Euro Spot
                                                 Exchange Rate
cpi_spain       float    %9.0g                 Consumer Price Index: All Items:
                                                 Total: Total for Spain
oil_price       float    %9.0g                 Spot Crude Oil Price: West Texas
                                                 Intermediate (WTI)
month           float    %tm                   Monthly date
ldollar_euro    float    %9.0g                 Log of dollar_euro
lcpi_spain      float    %9.0g                 Log of cpi_spain
loil_price      float    %9.0g                 Log of oil_price
--------------------------------------------------------------------------------
Sorted by: month
```

Exchange rate, cpi, and oil price



Source: Federal Reserve Economic Data (FRED)
Downloaded using -import fred- in Stata

Exchange rate, cpi, and oil price (12 month change)



Source: Federal Reserve Economic Data (FRED)
Downloaded using -import fred- in Stata

Conjugate prior with self-tightness equal to 1

```
 . matrix b0 = J(1,2,0)
. bayes,minnconjprior(mean(b0) selftight(1)) rseed(123) dryrun:         ///
>   var S12.lcpi_spain S12.ldollar_euro,exog(L(0 1)S12.loil_price) lags(1/2)
Model summary
------------------------------------------------------------------------------
Likelihood:
    S12_lcpi_spain
  S12_ldollar_euro ~ mvnormal(2,xb_S12_lcpi_spain,xb_S12_ldollar_euro,{Sigma,m
                     })
Priors:
{S12_lcpi_s~n:L(S12 2S12).lcpi_spain}                                     (1)
{S12_lcpi_s~n:L(S12 2S12).ldollar_~o}                                     (1)
   {S12_lcpi_s~n:L(1 S12).loil_price}                                     (1)
                 {S12_lcpi_s~n:_cons}                                     (1)
{S12_ldolla~o:L(S12 2S12).lcpi_spain}                                     (2)
{S12_ldolla~o:L(S12 2S12).ldollar_~o}                                     (2)
   {S12_ldolla~o:L(1 S12).loil_price}                                     (2)
                 {S12_ldolla~o:_cons} ~ varconjugate(2,2,3,b0,{Sigma,m},_Phi0)
                                        (2)
                            {Sigma,m} ~ iwishart(2,4,_Sigma0)
------------------------------------------------------------------------------
(1) Parameters are elements of the linear form xb_S12_lcpi_spain.
(2) Parameters are elements of the linear form xb_S12_ldollar_euro.
```

Conjugate prior / self-tightness equal to 1 (continue)

```
. bayes,minnconjprior(mean(b0) selftight(1))                          ///
>   noheader nomodelsummary rseed(123):                                ///
>   var S12.lcpi_spain S12.ldollar_euro,exog(L(0 1)S12.loil_price) lags(1/2)

Burn-in ...
Simulation ...
```

	Mean	Std. dev.	MCSE	Median	Equal-tailed [95% cred.	interval]
S12_lcpi_s~n						
lcpi_spain						
LS12.	.9785865	.0692543	.000693	.9792051	.8416478	1.113884
L2S12.	-.0260041	.0656856	.000649	-.0259195	-.1548474	.104381
ldollar_euro						
LS12.	-.0051642	.0089951	.00009	-.0052653	-.0227631	.0124788
L2S12.	-.0112962	.0091975	.000092	-.0111795	-.0290683	.0066003
loil_price						
S12.	.0163205	.0018713	.000019	.0163025	.0126649	.0201499
LS12.	-.0087195	.0019892	.000021	-.0087173	-.0125892	-.0048318
_cons	.0008534	.0004201	4.1e-06	.0008567	.0000322	.0016729
S12_ldolla~o						
lcpi_spain						
LS12.	-.4380218	.4976122	.004976	-.440278	-1.421946	.5452821
L2S12.	.4308779	.4713806	.004714	.4301426	-.4924678	1.36571
ldollar_euro						
LS12.	1.269213	.063328	.000633	1.268931	1.147171	1.392798
L2S12.	-.3679531	.0647047	.000647	-.3677778	-.4966777	-.2413958
loil_price						
S12.	.054238	.0133521	.000134	.0544479	.0277955	.0799516
LS12.	-.0504826	.0140335	.00014	-.0504495	-.0777651	-.0230531
_cons	-.0006596	.0029726	.00003	-.0006441	-.006569	.0050565
Sigma_1_1	.0000149	1.49e-06	1.5e-08	.0000148	.0000123	.0000181
Sigma_2_1	-.0000151	7.72e-06	7.7e-08	-.0000149	-.000031	-2.33e-07
Sigma_2_2	.0007681	.0000776	7.8e-07	.0007641	.00063	.0009322

Lag selection

- In the classical estimation we can select the optimal number of lags by using a few different information criteria like the ones implemented with `varsoc` (AIC, BIC, FPE, HQIC)

- In the Bayesian approach, we can perform the selection using posterior probabilities (with `bayestest model` and Bayes factors with `bayesstats ic`).

Lag selection

- Fit the competing models, saving the mcmc simulation and storing the results:

```
 . matrix b0 = J(1,2,0)
. bayes,minnconjprior(mean(b0) selftight(1)) rseed(123)              ///
> saving("$simul_dir\bvarcpi_sim1",replace):                         ///
> var s12.lcpi_spain s12.ldollar_euro,exog(L(0 1)S12.loil_price) lags(1/1)
. estimates store bvar1
. bayes,minnconjprior(mean(b0) selftight(1)) rseed(123)              ///
> saving("$simul_dir\bvarcpi_sim2",replace):                         ///
> var s12.lcpi_spain s12.ldollar_euro,exog(L(0 1)S12.loil_price) lags(1/2)
. estimates store bvar2
. bayes,minnconjprior(mean(b0) selftight(1)) rseed(123)              ///
> saving("$simul_dir\bvarcpi_sim3",replace):                         ///
> var s12.lcpi_spain s12.ldollar_euro,exog(L(0 1)S12.loil_price) lags(1/3)
. estimates store bvar3
. bayes,minnconjprior(mean(b0) selftight(1)) rseed(123)              ///
> saving("$simul_dir\bvarcpi_sim4",replace):                         ///
> var s12.lcpi_spain s12.ldollar_euro,exog(L(0 1)S12.loil_price) lags(1/4)
. estimates store bvar4
. bayes,minnconjprior(mean(b0) selftight(1)) rseed(123)              ///
> saving("$simul_dir\bvarcpi_sim5",replace):                         ///
> var s12.lcpi_spain s12.ldollar_euro,exog(L(0 1)S12.loil_price) lags(1/5)
. estimates store bvar5
.
. bayestest model bvar1 bvar2 bvar3 bvar4 bvar5
. bayesstats ic bvar1 bvar2 bvar3 bvar4 bvar5, basemodel(bvar5)
```

Lag selection

- Selection based on posterior probabilities and bayes factors:

```
. bayestest model bvar1 bvar2 bvar3 bvar4 bvar5
Bayesian model tests

             |   log(ML)       P(M)      P(M|y)
-------------+---------------------------------
       bvar1 | 1174.6430     0.2000     0.1301
       bvar2 | 1176.5430     0.2000     0.8699
       bvar3 | 1163.9431     0.2000     0.0000
       bvar4 | 1154.4924     0.2000     0.0000
       bvar5 | 1143.5097     0.2000     0.0000
-----------------------------------------------
Note: Marginal likelihood (ML) is computed using
      Laplace-Metropolis approximation.

.
.
. bayesstats ic bvar1 bvar2 bvar3 bvar4 bvar5, basemodel(bvar5)
Bayesian information criteria

             |       DIC     log(ML)     log(BF)
-------------+----------------------------------
       bvar1 |  -2447.45    1174.643    31.13335
       bvar2 | -2461.449    1176.543    33.03334
       bvar3 | -2443.685    1163.943    20.43347
       bvar4 | -2430.409    1154.492    10.98272
       bvar5 | -2413.784     1143.51           .
------------------------------------------------
Note: Marginal likelihood (ML) is computed
      using Laplace-Metropolis approximation.
```

Lag selection

- Change prior probabilities:

```
. bayestest model bvar1 bvar2 bvar3 bvar4 bvar5,  ///
>     prior(.3,.1,.35,.15,.1)
Bayesian model tests
```

	log(ML)	P(M)	P(M|y)
bvar1	1174.6430	0.3000	0.3097
bvar2	1176.5430	0.1000	0.6903
bvar3	1163.9431	0.3500	0.0000
bvar4	1154.4924	0.1500	0.0000
bvar5	1143.5097	0.1000	0.0000

```
Note: Marginal likelihood (ML) is computed using
      Laplace-Metropolis approximation.
```

- Compare bayes factors for the first two models

```
. bayesstats ic bvar1 bvar2,basemodel(bvar1)
Bayesian information criteria
```

	DIC	log(ML)	log(BF)
bvar1	-2447.45	1174.643	.
bvar2	-2461.449	1176.543	1.899989

```
Note: Marginal likelihood (ML) is computed
      using Laplace-Metropolis approximation.
```

Check stability condition

```
 . quietly bayes, minnconjprior(mean(b0) selftight(1)) rseed(123)         ///
>    saving("$simul_dir\bvar_sim1",replace):                               ///
>    var s12.lcpi_spain s12.ldollar_euro,exog(L(0 1)S12.loil_price) lags(1/2)

. estimates store ex_irf

. bayesvarstable

Eigenvalue stability condition                Companion matrix size =     4
                                              MCMC sample size      = 10000
```

Eigenvalue modulus	Mean	Std. dev.	MCSE	Median	Equal-tailed [95% cred.	interval]
1	.9494043	.0356101	.000356	.9508446	.8770518	1.013827
2	.8339488	.0863248	.000863	.8478731	.6392703	.9526843
3	.4197172	.1245125	.001245	.4076814	.2006145	.6890267
4	.0779642	.0654407	.000654	.0613823	.0025941	.2523488

```
Pr(eigenvalues lie inside the unit circle) = 0.9316
```

- The probability that all the eigenvalues are less than one supports the stability of the selected model.

Impulse-response function for interannual inflation

- Shocks on dollar-euro exchange rate

```
. quietly bayes, minnconjprior(mean(b0) selftight(1)) rseed(123)        ///
>   saving("$simul_dir\bvar_sim1",replace):                              ///
>   var s12.lcpi_spain s12.ldollar_euro,exog(L(0 1)S12.loil_price) lags(1/2)
. estimates store ex_irf
. bayesirf create order_er_ipc, step(40) set(order_er_ipc,replace)       ///
>    order(s12.ldollar_euro s12.lcpi_spain)
(file order_er_ipc.irf created)
(file order_er_ipc.irf now active)
(file order_er_ipc.irf updated)
. bayesirf graph oirf,impulse(S12.lcpi_spain S12.ldollar_euro)          ///
>                    response(S12.lcpi_spain)
```



Dynamic multiplier for interannual inflation

- Dynamic multiplier associated to changes in oil_price

```
. bayesirf graph dm,impulse(S12.loil_price) response(S12.lcpi_spain)   ///
>    ustep(10)
```



Forecasting

Bayesian forecasting

- Posterior predictive distribution of the replicated data:

$$\Pr(\mathbf{y}_{T+1:T+h}|\mathbf{y}_T) = \int f(\mathbf{y}_{T+1:T+h}|\mathbf{y}_T;\boldsymbol{\theta})p(\boldsymbol{\theta}|\mathbf{y})d\boldsymbol{\theta}$$

Where:

$$\begin{aligned}
\tilde{\mathbf{y}}^s_{T+1} &= \mathbf{y}_T + \mathbf{u}^1 &;\ \mathbf{u}^1 \sim N(0, \boldsymbol{\Sigma}^s) \\
\tilde{\mathbf{y}}^s_{T+2} &= \tilde{\mathbf{y}}^s_{T+1} + \mathbf{u}^2 &;\ \mathbf{u}^2 \sim N(0, \boldsymbol{\Sigma}^s) \\
\ldots \\
\tilde{\mathbf{y}}^s_{T+h} &= \tilde{\mathbf{y}}^s_{T+h-1} + \mathbf{u}^h &;\ \mathbf{u}^h \sim N(0, \boldsymbol{\Sigma}^s)
\end{aligned}$$

Save dynamic forecasts $(\tilde{\mathbf{y}}^s_{T+1}, \tilde{\mathbf{y}}^s_{T+2}, \ldots, \tilde{\mathbf{y}}^s_{T+h})$

Forecasting - `bayesfcast compute`

- Our final model contains an exogenous variable:

```
. quietly bayes, minnconjprior(mean(b0) selftight(1)) rseed(123)        ///
>    saving("$simul_dir\bvar_sim1",replace):                            ///
>    var s12.lcpi_spain s12.ldollar_euro,exog(L(0 1)S12.loil_price) lags(1/2)
```

- Get out of sample values for the exogenous variable loil_price.

```
. arima S12.loil_price,ar(1)  nolog
ARIMA regression
Sample: 2006m1 thru 2022m8                      Number of obs     =        200
                                                Wald chi2(1)      =    2069.62
Log likelihood = 85.34334                       Prob > chi2       =     0.0000

------------------------------------------------------------------------------
S12.         |                 OPG
  loil_price | Coefficient  std. err.      z    P>|z|     [95% conf. interval]
-------------+----------------------------------------------------------------
loil_price   |
       _cons |   .0585879   .1306455     0.45   0.654    -.1974726    .3146483
-------------+----------------------------------------------------------------
ARMA         |
          ar |
         L1. |   .9156011   .0201262    45.49   0.000     .8761545    .9550476
-------------+----------------------------------------------------------------
      /sigma |    .157204   .0043613    36.05   0.000     .1486561     .165752
------------------------------------------------------------------------------
Note: The test of the variance against zero is one sided, and the two-sided
      confidence interval is truncated at zero.
. predict loilp_hat2, y
(12 missing values generated)
. replace loil_price=loilp_hat if tin(2022m9,2022m12)
(4 real changes made)
```

Forecasting - bayesfcast compute

- Get dynamic forecasts and plot them (also save the simulated outcomes).

```
 . bayes, minnconjprior(mean(b0) selftight(1))          ///
>    rseed(123) saving("$simul_dir\bvar_sim1",replace): ///
>    var s12.lcpi_spain s12.ldollar_euro,               ///
>    exog(L(0 1)S12.loil_price) lags(1/2)
.
. bayesfcast compute bvar_, step(5)                           ///
>     mcmcsaving("$simul_dir\fcast_mcmc",replace)
. bayesfcast graph bvar_S12_lcpi_spain bvar_S12_ldollar_euro, ///
>     observed tlabel(749(2)755,angle(30))
```



Forecasting events

- How about obtaining probabilities for events associated to different levels of the endogenous variables?

```
. /* Use mcmc simulations for the predicted outcome variables */
. use "$simul_dir\fcast_mcmc",clear
. /* Rename to identify daily predictions */
. foreach name in S12_lcpi_spain S12_ldollar_euro {
.   forvalues i=752/755 {
.         local ti=`i´-751
.     rename `name´_`i´ `name´_`ti´
.   }
. }
. /* Separate events for t+1 to t+4 (Sep-Dec) */
. foreach var in inf_pos erate_pos inf_erate_pp inf_erate_pn {
.        generate `var´ = 0
. }
. forvalues i=1/4 {
.   gen      inf_pos_t`i´=cond(S12_lcpi_spain_`i´>0.1,1,0)
.   gen    erate_pos_t`i´=cond(S12_ldollar_euro_`i´>-0.15,1,0)
.   gen inf_erate_pp_t`i´=cond(S12_lcpi_spain_`i´>0.1 &          ///
>                           S12_ldollar_euro_`i´>-.1,1,0)
.   gen inf_erate_pn_t`i´=cond(S12_lcpi_spain_`i´>0.1 &          ///
>                           S12_ldollar_euro_`i´<-.1,1,0)
.   foreach var of varlist inf_pos erate_pos inf_erate_pp inf_erate_pn {
.     replace `var´ = `var´ + `var´_t`i´
.   }
. }
. foreach var of varlist inf_pos erate_pos inf_erate_pp inf_erate_pn {
.         generate `var´_sept_dec = cond(`var´== 4,1,0)
. }
```

Reporting event probabilities for 2022

- We can now combine `proportion` with `collect` to report probabilities for events associated to our forecasts:

```
 . collect clear
. forvalues i=1/4 {
.    collect inf_pos= e(b)[1,2] erate_pos = e(b)[1,4],              ///
>         tags(month[`i´]): proportion inf_pos_t`i´ erate_pos_t`i´
. }
. collect inf_pos=e(b)[1,2]  erate_pos = e(b)[1,4], tags(month[5]): ///
>      proportion inf_pos_sept_dec erate_pos_sept_dec
. foreach var of varlist inf_erate_pp inf_erate_pn {
.      collect `var´_sept_dec = e(b)[1,2], tags(month[5]):          ///
>              proportion `var´_sept_dec
. }
. quietly collect layout (result) (month)
. collect label levels month 1 "Sept" 2 "Oct" 3 "Nov" ///
>                           4 "Dec" 5 "Sept-Dec"
. collect label levels result                                       ///
>              inf_pos "Inflation>10%"                              ///
>              erate_pos "S12(Exc.rate)>-10%"                       ///
>              inf_erate_pp_sept_dec "Inf>10% & S12(Exc.rate)>-10%" ///
>              inf_erate_pn_sept_dec "Inf>10% & S12(Exc.rate)<-10%"
. collect style header result, title(label)
. collect style cell result, nformat(%6.2f)
. collect style column, extraspace(1)
```

Reporting event probabilities for 2022

```
. collect preview
```

	Sept	Oct	Nov	Dec	Sept-Dec
Result					
Inflation>10%	0.66	0.64	0.62	0.59	0.46
S12(Exc.rate)>-10%	0.64	0.69	0.73	0.75	0.55
Inf>10% & S12(Exc.rate)>-10%					0.04
Inf>10% & S12(Exc.rate)<-10%					0.27

Summing up

- Frequentist analysis base the conclusions on the distributions of statistics derived from random samples, assuming unknown fixed parameters.
- Bayesian analysis answers questions based on the distribution of parameters conditional on the observed sample.
- Bayesian VAR models are particularly convenient when working with small samples. Shrinking the parameter space with the priors allows controlling the number of lags more effectively.
- Impulse-response analysis and Forecasting are based on the full probability distributions for the parameters and the predictions.
- The posterior predictive distribution can be used to define events that can be evaluated for policy analysis.

References

- Doan, T., R. B. Litterman, and C. Sims. 1984. *Forecasting and conditional projection using realistic prior distributions.* Econometric Reviews 1: 1–100.
- Garratt, A., Lee, K., M.H. Pesaran and Y. Shin. 2006, *Global and National Macroeconometric Modeling: A long-run structural approach.* Oxford University Press, Oxford.
- Litterman, R. B. 1986. *Forecasting with Bayesian vector autoregressions-Five years of experience.* Journal of Business and Economic Statistics 4: 25–38.
- Sanchez, G. and Zavarce 2013, *Prospectos para la economía: Una aplicación con modelos VAR cointegrados y proyecciones probabilísticas.* in "Global and National Macroeconometric Modeling" Stata Press.